

Direct Abstraction 15 User Manual

Matthew Kaes

Table of Contents

Using Direct Abstraction:	4
Running Code:	4
Direct Abstraction IDE:	4
Building Code in the IDE:	4
Syntax:	5
Comments:	5
Base types:	5
Nil:	6
Integer:	6
Double:	6
Strings:	6
Containers:	7
Reference:	8
Symbols:	9
Variables:	9
Globals:	10
Attributes:	10
Control Statements:	11
If Statement:	11
Else Statement:	11
Scoping:	12
While statements:	13
For statements:	13
Declarations:	13
Class Declarations:	14
Function Declarations:	14
Operators:	15
Addition - Subtraction:	15
Multiplication – Division:	16
Built in functions:	16
print:	16
print_color:	16
get_value:	17

get_symbol:.....17

Using Direct Abstraction:

Direct Abstraction is based on two parts; the interpreter which parses your code and executes your program, and the abstraction library which informs the interpreter how to perform the required operations. In most cases the interpreter and the abstraction library are built into one binary file, but this is not always the case as the interpreter can have added functionality with the use of an extension library in the form of a .DLL file. To run the interpreter simply have a file named "main.txt" which has your code in plain text format in the same directory as the interpreter and the abstraction library. A new instance of the interpreter will launch and your code will be loaded into memory. For DOS support direct abstraction can run natively from the command prompt and can also run natively in Cygwin.

Running Code:

In order for your code to run it must have a valid entry point. Depending on your version of DA (direct abstraction) your code will have different entry points. For version of DA older than 11 the entry point for the code will be the function main. The smallest executable code for DA 11 and above is as follows.

```
1 #Filename Main
2
3 def main
4   #code goes here.
5 end
```

Direct Abstraction IDE:

Direct Abstraction comes with a dedicated IDE (Integrated Development Environment) on the Windows operating system. From the Direct Abstraction IDE you can run all of your code with a click of a button, choose which version of DA you wish to run under, and manage your code in a simple file like system. The IDE also comes with features such as syntax highlighting and will embed the require headers for your code.

Building Code in the IDE:

When you first start up the IDE you will be met with a stub project that includes a test Main script. From here you add to the project to run your own code or add new scripts to the project with the "Add Script" button. The text box above this button will make the name for the new script that you add to your project. This name is not permanent however and can be changed at any time by modifying the "#Filename" declaration at the start of your file. Once you are happy with your code you can run it at any time from the IDE by pressing the "Test Code" button. Testing code does not lock up the IDE and

you are able to spawn multiple test session at the same time by clicking the button multiple times. You can also change what version of DA you are using by changing the drop down box next to the “Test Code” button.

Syntax:

Direct Abstraction has syntax that is similar to most higher level languages such as Ruby and Python. Direct Abstraction supports weak typing like most of these languages as well so all types are inferred at runtime.

Comments:

All comments in Direct Abstraction use the “#” character. All characters after the “#” character are considered comments and will be stripped out at run time. Directives are also a type of comment at the start of the file. It is also possible to use a comment mid line by placing a second “#” to terminate the comment; however this is not supported in all versions of DA.

```
1 #Filename Main
2
3 def main
4   #Example Comment
5
6   x = 4 #comment at the end of a line.
7 end
```

Base types:

Direct Abstraction has several base data types. From these base types all other types are built on. Since in DA you have no control over implicit casting DA follows the “lossless” implicit cast. This means that base types will only go up to higher information types but never down. All the base times from lowest order to highest order are; Nil, Integer, Double, String, Container, Reference. From this order we can easily derive the rules for implicit casting. If an operation takes in an Integer and a Double the Integer will be converted to a Double and the operation will be performed however the Double will never be converted to an Integer. You can force a type conversion (explicit casting) by using the “Convert” function. (See “Built In Library”)

Nil:

Nil is any non-initialized data. Nil converts into zero when cast into an Integer, 0.0 when cast into a Double, "nil" when cast to a string, and 'empty' when cast into a container. "nil" is also a registered keyword and can be used to create and return nil whenever you desire. Nil can also be a byproduct of a function or an operation that didn't have valid arguments such as "nil + 4" or ""string" / 6".

```
1 #Filename Main
2
3 def main
4   #invalid: results in nil
5   4 + nil
6
7   #valid: results in the string "nil data"
8   nil + " data"
9 end
```

Integer:

Integer is the lowest order base type to hold useful information. The integer type holds onto all whole numbers from $-(2^{63})$ to $2^{63} - 1$. Integers are similar to the "long long int" data type in C/C++.

Double:

Doubles hold onto and deal with all floating point arithmetic. Doubles take the form of 0.0 where the decimal place is required. If a double is a whole number such as 2 the form of the double must be 2.0. This resolves issues that could be mistaken for the dot operator (.) and also makes sure that it is differentiable from an integer.

Strings:

Strings are created by putting any line of code in quotes (" "). Strings are not limited in any size and will grow in memory accordingly in order to match the size required for them. You can manipulate strings by using the addition operator (+) to append to the string. Strings also support a small list of special escape characters such as "\n" which will replace the two characters with the newline character at runtime.

```
1 #Filename Main
2
3 def main
4   #Prints "Hello World"
5   var = "Hello"
6   var = var + " World"
7   print(var)
8
9   #Also prints "Hello World"
10  print("Hello" + " World")
11
12  #Prints "GPA: 3.4"
13  print("GPA:" + 3.4)
14 end
```

Containers:

Containers hold any number of objects from Integers, nil objects, doubles, strings, and even other containers. Containers can hold either a single type of object, or any number of objects. Containers in this sense are not limited in size or types held. When constructing containers the container will always be allocated and release whatever memory is required in order for the container to operate. Containers have two operators that can be used on them including the bracket operator ([,]) and the addition operator (+) that appends an element to the end of the container. When using the addition operator on two arrays the second container is instead concatenated to the first container and a new container composed of both sets is returned. Because Direct Abstraction always assures that there is memory for an array a symbol will be created into a container if the bracket operator is ever used on it and the maximum required index for the container will become its new size. Even if a new value isn't assigned to the memory location the container will still grow. If a container grows and it has no data to fill with all of the empty space will be padded with "nil" data. Since containers always assure that the index you are accessing is valid you can use the "nil" keyword as an index to create an empty set.

```
1 #Filename Main
2
3 def main
4   #Creates a container of [5]
5   var[0] = 5
6
7   #Creates a container of [5, 6]
8   var = var + 6
9
10  #Creates a container of [nil, nil, "Str"]
11  var2[3] = "Str"
12
13  #Prints Array: [nil, nil, "Str", 5, 6]
14  print(var2 + var)
15
16  #Creates an empty container
17  var3[nil]
18
19  #Prints Array: EMPTY
20  print(var3)
21 end
```

Reference:

A reference is created only in two instances, when two variables are coupled with the couple operator (":") and when a dynamic object is created with the "new" keyword. References act like the object and can be assigned to and used as if it was that object but is technically a different symbol. References allow for objects to be passed to functions in such a way that the function can affect an external value rather than relying solely on return values.

```
1 #Filename Main
2
3 def main
4   #set a variable with a value of 2
5   var = 2
6
7   #Make another variable couple the previous variable
8   var2 : var
9
10  #Set the value of var2
11  #var now prints as 8
12  var2 = 8
13  print(var)
14
15  #decouple var2
16  var2 : nil
17
18  #Set the value of var2
19  #var still prints as 8
20  var2 = "str"
21  print(var)
22 end
```


Symbols:

There are three different types of symbols that you can use to store all of your information in and perform your operators on. The three types of operators are variables, globals, and attributes. These three types of symbols are all the same and have the exact same behavior and the only thing that differs between them is the scope in which the symbol exists. Variables are function scoped and exist for the entirety of the function. Globals will remain in effect for the entirety of the program. Attributes are special and are dynamic object scoped and will remain for as long as the dynamic object that they are attached to remains in working memory. This means that attributes will be tied to another symbol be it a variable, a global, or another attribute.

Variables:

Variables take the form of any regular word that is not a keyword or a registered word (such as a function or class made by the user). A variable must start with a readable character and can contain numbers, underscores, and other characters. Variables are function scoped which means that in the function that they are declared in they are useable in all places. Since variables are function scoped, all variables used in a function are constructed when the function is called.

```
1 #Filename Main
2 def main
3   #create a variable
4   var
5
6   #print out the variable
7   print(var)
8
9   #print out another variable
10  print(var2)
11
12  #Variables are valid in all places
13  #In this case "value" will be printed
14  if 1
15    var3 = "value"
16  end
17  print(var3)
18 end
```

Globals:

Globals have all of the same syntax requirements as variables but require a “\$” at the start of the variable name. Globals can be created anywhere and at any time.

```
1 #Filename Main
2 def main
3   #create a global
4   $var
5
6   #print out the global variable
7   print( $var )
8
9   #Make the global in another function
10  #The following print will print 4
11  initialize_global
12  print( $var2 )
13 end
14
15 #A function that creates a global
16 def initialize_global
17   $var2 = 4
18 end
```

Attributes:

Attributes have all of the same syntax requirements as variables but require a “@” at the start of the variable name. Unlike globals and variables, attributes can only be created inside of a class object. Attributes cannot be created anywhere else in the source code.

```
1 #Filename Main
2 def main
3   #create an object and print
4   #prints CLASSOBJ: obj { @x = 4, @y = 8 }
5   print( new obj )
6
7   #Attributes only allowed in classes
8   @var = 5 #This will create an error
9 end
10
11 #A classtype object that contains attributes
12 class obj
13   def initialize
14     #Creates two Attributes, x and y
15     @x = 4
16     @y = 8
17   end
18 end
```

Control Statements:

There are three different types of control statements that are used in Direct Abstraction. Control statements are used to control the flow of the program and control logic. Control statements fall into two simple categories, conditionals and loops. Conditionals execute a set of instructions if a certain condition is validated. If the condition isn't validated then the code will not be executed. Loops are used to rerun code multiple times. The number of times the code is run and the exit requirements for a loop are based on the type of loop used.

If Statement:

If statement is a conditional that executes if a statement is valid. A valid statement is any statement that evaluates as "true" when "convert(statement, "boolean")" is used on it. Nil statements are considered not valid and are not executed.

```
1 #Filename Main
2 def main
3   #get user input and store it in x
4   x = get_value()
5
6   #set a value to guess
7   val = 20
8
9   #check if the user input was
10  #less than val
11  if( x < val)
12    #print out a response
13    print("Your number is less than " + val)
14  end
15 end
```

Else Statement:

"Else" statements can be added to the end of an "if" statement. If the "if" statement is passed an invalid value then the else statement will be executed instead. "Else" statements can only be used in conjunction with "if" statements.

```
1 #Filename Main
2 def main
3   #get user input and store it in x
4   x = get_value()
5
6   #set a value to guess
7   val = 20
8
9   #check if the user input was
10  #less than val
11  if( x < val)
12    #print out a response
13    print("Your number is less than " + val)
14  else
15    #print out a response
16    print("Your number is greater than " + val)
17  end
18 end
```

Scoping:

Every time a control statement is used it creates its own scope. Scope is defined as blocks of code to be executed inside of control statements. Unlike languages like C and C++ scope does not affect variable declarations.

```
1 #Filename Main
2 def main
3   #get user input and store it in x
4   x = get_value()
5
6   #set a value to guess
7   val = 20
8
9   #check if the user input was
10  #less than val
11  if( x < val)
12    #print out a response
13    print("Your number is less than " + val)
14  else
15    #add a control statement in the scope
16    if( x == val)
17      #print out a response
18      print("Your number is equal too " + val)
19    else
20      #print out a response
21      print("Your number is greater than " + val)
22    end
23  end
24 end
```

While statements:

While statements are a loop that will continue to execute as long as the statement it qualifies is valid. While loops are good at controlling repeated code and setting up persistent programs.

```
1 #Filename Main
2 def main
3   #italize a variable x
4   x = 0
5
6   #get user input till they input something
7   #larger then or equal to 20.
8   print("Enter a number greater than 20")
9   while(x < 20)
10    x = get_value()
11  end
12 end
```

For statements:

For loops are used to loop over all the values in a container. A variable is given to act as the element to be used as a reference to each element in the container. If an empty container or a symbol that is not a container then that means the input is considered not valid and the loop is not executed.

```
1 #Filename Main
2 def main
3   #italize a container
4   x[nil]
5
6   #Fill x with values
7   x = x + 2
8   x = x + 3
9   x = x + 12
10
11  #double each value in x
12  for value in x
13    value = value * 2
14  end
15
16  #prints ARRAY: {4, 6, 24}
17  print(x)
18 end
```

Declarations:

Declarations are used to inform Direct Abstraction how to handle symbols as objects and how to call functions in memory. There are two type of declarations used in Direct Abstraction.

Class Declarations:

The first kind of declaration is class declarations. Class declarations tell Direct Abstraction what symbols should be considered objects instead of variables. Objects can have their own functions embedded in them and are able to have persistent values inside of them known as attributes.

```
1 #Filename Main
2 def main
3   #makes a new player object
4   actor = new player
5
6   #calls actor's move function
7   actor.move()
8 end
9
10 #create the class called player
11 class player
12   def initialize
13     #initialize x and y attributes
14     @x = 0
15     @y = 0
16   end
17   def move
18     #increment the x and y values by 1
19     @x += 1
20     @y += 1
21   end
22 end
```

Function Declarations:

Function declarations tell Direct abstraction that the following symbol is a function and maps to a segment of code that should be called. At the end of a function declaration should be the arguments that are to be passed to the function.

```
1 #Filename Main
2 def main
3   #Prints 4, 8.0, 12, and 24
4   print(double(2))
5   print(double(4.0))
6   print(double("6"))
7   print(double(double("6")))
8 end
9
10 #make a function that doubles a passed value
11 def double(value)
12   return value * 2
13 end
```

Operators:

Direct Abstraction has support for a number of operators to preform math and other features. The order in which operators get executed is based on the priority of the operation. Operations with higher priorities are executed first followed by lower priority operations. If two operations with the same priority are on the same line then they will be executed from left to right.

List of all Direct Abstraction operation priorities

Symbol	Priority	Symbol	Priority
Object	11	Addition / Subtraction	5
Range Operator	10	Modulo	4
Array Bracket	9	Dot Operator	4
Function	9	Compare	3
Couple Operator	8	Greater/Less Than	3
Power	7	Assignment	2
Mul / Division	6	Keyword	1

Addition - Subtraction:

Addition and subtraction are two of the most basic arithmetic functions. Both have a priority of 5. It should be noted that addition is overloaded to have special functionality other than just arithmetic. When used on strings addition concatenates two strings together making a longer string. When used on containers addition adds the element to the container.

```

1 #Filename Main
2 def main
3   #Preform simple addition and subtraction
4   x = 2 + 6 - 4
5   #Prints 4
6   print(x)
7   y = 10 + x
8   #Prints 14
9   print(y)
10
11  #prints "Y is set to 14"
12  print("Y is set to " + y)
13
14  array[nil]
15  array = array + 2
16  array = array + "str"
17  array = array + true
18
19  #prints ARRAY: {2, "str", true}
20  print(array)
21 end

```

Multiplication - Division:

Multiplication and division perform the same functionality as they do in other languages. Both have a priority value of 6. With division if both values are numbers then integer division is performed and the returned value is the floor of the division.

```
1 #Filename Main
2 def main
3   #Preform simple multiplication and division
4   #notice the integer division
5   value = 2 * 6 / 4
6   #prints 2
7   print(value)
8
9   value = 2 * 6 / 4.0
10  #prints 3
11  print(value)
12
13  #Creates an error at runtime
14  value = 10 / 0
15 end
```

Built in functions:

Direct Abstraction has a number of built in functionality. Since though Direct Abstraction it is impossible to gain access to low level functionality it has been supplied for the end user.

print:

Print takes a single argument and prints it out onto the screen in

```
1 #Filename Main
2 def main
3   #prints "Hello World"
4   print("Hello World")
5 end
```

white.

print_color:

Print_color takes two arguments. The first argument is the symbol to be printed. The second dictates the color.

```
1 #Filename Main
2 def main
3   #prints "Hello World" in dark red
4   print_color("Hello World", 4)
5 end
```


get_value:

Get_value takes no arguments. Get_value gets a number or a double from stdin.

```
1 #Filename Main
2 def main
3   #Gets a number or double from the user
4   x = get_value()
5 end
```

get_symbol:

Get_symbol takes no arguments. Get_symbol a string from stdin.

```
1 #Filename Main
2 def main
3   #Gets a string from the user
4   x = get_symbol()
5 end
```